GESS — general linear system solution with condition estimation

Purpose: GESS (GEneral System Solution) solves the system AX = B where A is a general matrix. An estimate of the condition of A is provided.

Usage: CALL GESS (N, A, IA, B, IB, NB, COND)

N	$\longrightarrow$	the number of equations
A	$\longrightarrow$	the array, dimensioned (IA, KA) in the calling program, where IA ≥ N and KA ≥ N, containing the N × N coefficient matrix A is overwritten during the solution.
IA	$\longrightarrow$	the row (leading) dimension of A, as dimensioned in the calling program
B	$\longrightarrow$	the matrix of right-hand sides, dimensioned (IB, KB) in the calling program, where IB ≥ N and KB ≥ NB
	$\longleftarrow$	the solution X
IB	$\longrightarrow$	the row (leading) dimension of B, as dimensioned in the calling program
NB	$\longrightarrow$	the number of right-hand sides
COND	$\longleftarrow$	an estimate of the condition number of A (see **Note 1**)

Note 1: The condition number measures the sensitivity of the solution of a linear system to errors in the matrix and in the right-hand side. If the elements of the matrix and the right-hand side(s) of your linear system have **d** decimal digits of precision, the solution might have as few as **d** $- \log_{10}$ (COND) correct decimal digits. Thus if COND is greater than 10^{BdP}, there may be no correct digits.

If the given matrix, A, is known in advance to be well-conditioned, then the user may wish to use the routine GELE, which is a little faster than GESS. Ordinarily, however, the user is strongly urged to choose GESS, and to follow it by a test of the condition estimate.

Note 2: Users who wish to solve a sequence of problems with the same coefficient matrix, but different right-hand sides *not all known in advance*, should not use GESS, but should call subprograms GECE, GEFS and GEBS. (See the example of GEDC.) GECE is called once to get the LU decomposition (see the introduction to this chapter) and then the pair, GEFS (forward solve) and GEBS (back solve), is called for each new right-hand side.

Error situations: *(The user can elect to 'recover' from those errors marked with an asterisk — see *Error Handling*, Framework Chapter)

Number	Error
1	N < 1
2	IA < N
3	IB < N
4	NB < 1
10 + k*	singular matrix whose rank is at least k

Double-precision version: DGESS with A, B, and COND declared double precision

Complex version: CGESS with A and B declared complex

Storage: N integer locations and
N real (double precision for DGESS, complex for CGESS) locations of scratch storage in the dynamic storage stack

Time:

$$\frac{N^3}{3} + N^2 \times (\frac{9}{2} + NB) + N \times (\frac{19}{6} + NB) \text{ additions}$$

$$\frac{N^3}{3} + N^2 \times (\frac{5}{2} + NB) + N \times (\frac{7}{6} + NB) \text{ multiplications}$$

$$\frac{N^2}{2} + N \times (\frac{3}{2} + NB) \text{ divisions}$$

Method: Gaussian elimination with partial pivoting.
See the reference below for the method used to estimate the condition number.
GESS calls GECE, GEFS, and GEBS.

See also: GEBS, GECE, GEDC, GEFS, GELE, GELU

Authors: Linda Kaufman and Doris Ryan

Reference: Cline, A. K., Moler, C. B., Stewart, G. W., and Wilkinson, J. H., An estimate for the condition number, *SIAM J. Numer. Anal. 16* (1979), 368-375.

Example: The following program solves a 5×5 system with two right-hand sides

```
      INTEGER N, IREAD, I1MACH, I, NB, IWRITE, J
      REAL A(5,5), B(5,2), COND
      N=5
      IREAD=I1MACH(1)
C
      DO 10 I=1,N
          READ(IREAD,1) (A(I,J),J=1,N)
   1      FORMAT(1X,5F10.0)
  10  CONTINUE
C
      NB=2
      DO 20 I=1,N
          READ(IREAD,11) (B(I,J),J=1,NB)
  11      FORMAT(1X,2F10.3)
  20  CONTINUE
C
C SOLVE AX = B  BY CALLING GESS
C
      CALL GESS(N,A,N,B,N,NB,COND)
      IWRITE=I1MACH(2)
      WRITE(IWRITE,21) COND
  21  FORMAT(52H AN ESTIMATE OF THE CONDITION NUMBER OF THE MATRIX =,
     1         E14.7)
C
      WRITE(IWRITE,22)
  22  FORMAT(27H THE COMPUTED SOLUTION X IS,//)
      DO 30 I=1,N
          WRITE(IWRITE,23) (B(I,J),J=1,NB)
  23      FORMAT(1H,5F20.7)
  30  CONTINUE
C
      STOP
      END
```

For the input matrix given by:

1.	−2.	3.	7.	−9.
−2.	8.	−6.	9.	50.
11.	−6.	18.	−15.	−18.
7.	2.	−15.	273.	173.
−9.	50.	−18.	6.	1667.

and the following right-hand sides:

30.	29.419
−191.	−190.994
133.	133.072
−986.	−985.775
−6496.	−6495.553

the following results were obtained on the Honeywell 6000 computer at Bell Labs:

```
AN ESTIMATE OF THE CONDITION NUMBER OF THE MATRIX = 0.2759414E 04
THE COMPUTED SOLUTION X IS
```

2.0000004	2.4800003
4.9999970	4.8709986
2.9999988	2.6439993
−1.0000001	−1.0320001
−3.9999999	−3.9970000

The true solution to this problem is:

2.	2.48
5.	4.871
3.	2.644
−1.	−1.032
−4.	−3.997

Notice that a seemingly slight change in the right-hand side causes the solution to change noticeably. Furthermore, the relative error in the solution is about 2×10^{-7}. On the Honeywell computer, which has about 8 decimal digits for single-precision numbers, this represents the loss of about 1.5 decimal digits. A loss of up to 2×10^{-5} could be expected in light of the analysis given below.

Let Δb represent a perturbation in the right-hand side of a linear system.
If $Ax = b$ then

$$A(x + \Delta x) = b + \Delta b$$

where

$$\frac{||\Delta x||}{||x||} \le K(A) \left[\frac{||\Delta b||}{||b||} \right]$$

where $K(A)$ is the condition number of A, $K(A) = ||A||\,||A^{-1}||$ and $||\cdot||$, is some norm, e.g., $||x||_1 = \sum_{i=1}^{n} |x_i|$ if x is a vector.

The methods used in our linear equation package are guaranteed to provide an accurate answer to a slightly perturbed problem. If we assume that our method produces the correct answer to a problem where $||\Delta b|| \le \varepsilon ||b||$, where ε is the machine precision, then on the Honeywell 6000 where ε is about 10^{-8}, a relative error for the above example of 2×10^{-5} would not be surprising.

In our example one may consider the first column of B as b in (1.1), and the second column of B as $b + \Delta b$, so that $||\Delta b|| / ||b||$ is approximately .00015 using the $||\cdot||_1$ norm. If we look at the second solution as $x + \Delta x$ in (1.1) and the first solution as x, then $||\Delta x|| / ||x||$ is approximately .07. Thus equation (1.1) indicates that the condition number is at least 400, and the condition estimate, which at first appeared to be conservative, was in fact quite realistic.